as well as helping to co-ordinate research, consolidate existing knowledge into a single unit and identify research needs. However, their complexity, the research required to construct them and their tendency to lack generality (Jørgensen, 1983b) restricts their availability for use in predicting fish yield in developing and under-developed countries. In such countries there is a need to manage existing fisheries and to develop new fisheries to make maximum use of the available resources. In such cases, rapid, order of magnitude estimates are generally more important than precision (Ryder, 1978). The multi-species model, currently in a crude, unvalidated form, is designed as a first step in compromising between optimal complexity and minimum data requirements. The approach adopted was largely empirical and the refinements necessary to improve the model could also be derived empirically. The improvements to the model most urgently required are:

(a) a validated relationship between fish yield and some relatively easily obtained predictor or predictors for southern African impoundments. Potential predictors are the morphoedaphic index, phosphorus or primary production. At present estimates of fish yield are the least common of these parameters, particularly in South Africa.

(b) relationships including temperature, turbidity, latitude and water quality (e.g. pH, ammonia, dissolved oxygen, algal assemblages) which will permit prediction of at least approximate fish species community composition.

(c) the response of different fish species to exploitation and the response of other fish species in the system to exploitation of one or more populations. These relationships will undoubtably be the most difficult to describe.

The multi-species model has used the OECD (1982) relationships and the Oglesby (1977a) relationship to predict yield. These were validated for predicting fish yield in Hartbeespoort Dam under current conditions (Section 7.4.2) but not for other water-bodies and it is likely that the predictions would be invalid if the phosphorus concentration of Hartbeespoort Dam was reduced to 100 $\mu g.\ell^{-1}$ (Section 7.5.2). Therefore, these relationships need to be validated, improved or replaced in the multi-species model.

New impoundments are continually being constructed in southern Africa and in most parts of the world. In developing countries urbanisation, industrial development and natural climatic changes can alter the nature of a water body. The ability to make optimal use of the fish resources in these countries would be improved by the ability to predict the likely fish species composition of new or changing water bodies. Such relationships could be used to identify potential target species and hence assess the fishing gear requirements, potential market for the catch, and viability of the fishery. They could also be used to assess the impact of a proposed change, such as release of effluent into a lake, on the fish resource.

Most African freshwater fisheries are multi-species fisheries (e.g. Lake Kariba - Marshall *et al.*, 1982; Lake McIlwaine - Marshall, 1982; Lake Tanganyika - Coulter, 1981 and Hartbeespoort Dam - this study). The problem of managing multi-species fisheries is probably the central issue facing fisheries science at present (Cushing, 1981). The problem of species interactions was addressed in the multi-species model, but at a very simple level by assuming exact compensation for changes in yield in one species by the remaining species. This concept might be close to reality in Hartbeespoort Dam at present (Section 7.5.1), but is unlikely to be adequate to describe the true situation in a community of species occupying a variety of trophic levels. The problems of modelling or predicting such interactions are large. The apparent relationship between *C. gariepinus* and potential competitors, which was suggested to be one where *C. gariepinus* could not compete with other species, but was able to make use of under-utilised food resources, could be described empirically (Section 6.4.1) and this type of relationship could be incorporated in the multi-species model. However, for this, a similar relationship would have to be described between *C. gariepinus* yield and species diversity.

The problem of species interactions in yield models could be approached empirically, as described for *C. gariepinus*, but the data do not exist, at least in southern Africa, for derivation of significant relationships. Alternatively it could be approached from a theoretical viewpoint considering energy transfer through trophic levels (e.g. Kerr, 1974). In practice it will probably be found that a combination of the two approaches is required and that theoretical concepts will be modified or supplemented by empirical information gained from changes in fish species composition observed in existing fisheries.

Attempts to solve the many problems associated with multi-species modelling may lead towards increasing complexity. However, in attempting to improve the multi-species model or to build a similar general model, it is necessary to set the objectives, the levels of precision and generality required, and to use the minimum complexity necessary to achieve these objectives. Unnecessary complexity may only serve to increase potential error by increasing the number of incompletely or poorly described relationships in the model (Reckhow & Chapra, 1983).

In conclusion, the study leading to the construction of the single-species models has significantly increased the available knowledge on the nature of the Hartbeespoort Dam fish community and factors controlling its surplus production. Much of this knowledge will be applicable to other systems. The models themselves provide useful tools, even in their present incomplete forms, to assist in management of the fishery. The single-species models predicted a mean total yield of 285 $kg.ha^{-1}.yr^{-1}$ from Hartbeespoort Dam under current conditions and showed that the fish community, particularly *C. carpio*, was being over-exploited. Yields from other African water bodies ranged from 4 to 500 $kg.ha^{-1}.yr^{-1}$, with yields from only one of the 32 localities recorded exceeding the predicted yield from Hartbeespoort Dam (Frye & Iles, 1972). The hypertrophic status of Hartbeespoort Dam was primarily responsible for the high yield and also influenced the species composition of the fish community during the period of study. However, a reduction of the lake phosphorus concentration to less than 100 $\mu g.\ell^{-1}$ would be necessary to induce substantial changes in the yield and the species composition of the community. In contrast, the potential yield of the system was sensitive to changes in the volume of the lake and the winter temperatures. The shared food items in the diet of juveniles of the three major species indicated competition for resources, and thus changes in the standing stock of one of these species, would generate a response in the stocks and hence yield of the remaining species. Therefore, this study demonstrated the need to consider the nature of the fish populations, their interrelationships and their interaction with the environment, in the comprehensive management of an intensive fishery. The expansion and validation of the multi-species model for southern African freshwaters would provide a tool to facilitate management for optimal yield and the development of fisheries in this region.

Attempts to solve the many problems associated with multi-species modelling may lead towards increasing complexity. However, in attempting to improve the multi-species model or to build a similar general model, it is necessary to set the objectives, the levels of precision and generality required, and to use the minimum complexity necessary to achieve these objectives. Unnecessary complexity may only serve to increase potential error by increasing the number of incompletely or poorly described relationships in the model (Reckhow & Chapra, 1983).

In conclusion, the study leading to the construction of the single-species models has significantly increased the available knowledge on the nature of the Hartbeespoort Dam fish community and factors controlling its surplus production. Much of this knowledge will be applicable to other systems. The models themselves provide useful tools, even in their present incomplete forms, to assist in management of the fishery. The single-species models predicted a mean total yield of 285 $kg.ha^{-1}.yr^{-1}$ from Hartbeespoort Dam under current conditions and showed that the fish community, particularly *C. carpio*, was being over-exploited. Yields from other African water bodies ranged from 4 to 500 $kg.ha^{-1}.yr^{-1}$, with yields from only one of the 32 localities recorded exceeding the predicted yield from Hartbeespoort Dam (Fryer & Iles, 1972). The hypertrophic status of Hartbeespoort Dam was primarily responsible for the high yield and also influenced the species composition of the fish community during the period of study. However, a reduction of the lake phosphorus concentration to less than 100 $\mu g.\ell^{-1}$ would be necessary to induce substantial changes in the yield and the species composition of the community. In contrast, the potential yield of the system was sensitive to changes in the volume of the lake and the winter temperatures. The shared food items in the diet of juveniles of the three major species indicated competition for resources, and thus changes in the standing stock of one of these species, would generate a response in the stocks and hence yield of the remaining species. Therefore, this study demonstrated the need to consider the nature of the fish populations, their interrelationships and their interaction with the environment, in the comprehensive management of an intensive fishery. The expansion and validation of the multi-species model for southern African freshwaters would provide a tool to facilitate management for optimal yield and the development of fisheries in this region.

APPENDIX I DISCRIMINANT ANALYSIS

(a) Terms used in discriminant analysis program (DISCLAR) to determine age class strength from length frequencies.

All standard length parameters refer to the transformed SL unless otherwise stated. In the final analysis the transformation was SL^2.

Term	Definition
AMAX	Maximum discriminant function.
DIFFER	Total number of changes in age class classification after each iteration.
DLOOPS	Number of iterations.
PROP	Expected proportion of total in each age group.
SCR	No. of spinal rings.
SEX	Sex of each fish (1 = male, 2 = female).
SL	Standard length of each fish.
SLMEAN	Mean SL of age group - unclassified sample (UC).
SUM	Sum of SL of age class (UC).
SUMSQ	Sum of squares of standard length of age class (UC).
TL	Total length of fish.
TMR	Mean SL classified samples (CS).
TN	No. of fish in age class (CS).
TNUM	No. of fish in age class (UC).
TNUM1	No. of fish assigned to age class in previous iteration.
TN1	No. of fish in age class - (UC) + (CS) (TOT).
TOTNUM	Total number of fish (UC).
TR	Sum of SL of age class (CS).
TRSUM	Untransformed sum of SL of age class (UC).
TR1	Sum of SL of age class (TOT).
TSR	Sum of squares of SL of age class (CS).
TSR1	Sum of squares of age class (TOT).
TSUM	(Total number of fish (UC)) - (No. of age classes).
TTSR	Total spine radius.
VARR	Variance of SL of age class (TOT).
VARRP	Pooled variance for all age classes (TOT).

Appendix 1b. The FORTRAN V Programme, DISCLAR.

```
C      A PROGRAM TO CALCULATE THE MEAN AND VARIANCE OF CLARIAS
C      LENGTH AT AGE FROM DATA ON CLAR. FROM THIS INFORMATION
C      CLARIAS OF UNKNOWN  GE CAN BE ASSIGNED TO AGE CLASSES ON
C      THE BASIS OF LENGTH BY DISCRIMINANT ANALYSIS.
       PROGRAM DISCLAR(CLAR,CLAR22,UIT,TAPE1=CLAR,TAPE2=
      2CLAR22,TAPE6=UIT,OUTPUT)
       DIMENSION R(10),TR(10),TSR(10),TN(10),TMR(10),VARR(10)
      A,SUM(10),TRSUM(10),SUMSQ(10),TNUM(10),SLMEAN(10),PROP(10),V(10)
      B,TNUM1(10),TR1(10),TSR1(10),TN1(10)
       DLOOPS=0
       DO 4 AA=1,10
          R(AA)=TR(AA)=TSR(AA)=TN(AA)=TMR(AA)=VARR(AA)=0
          SUM(AA)=TRSUM(AA)=SUMSQ(AA)=TNUM(AA)=TNUM1(AA)=0
          PROP(AA)=0.125
     4 CONTINUE
       DO 30 I=1,91
       READ (1,5,END=30)SL,SEX,SCR,TTSR,R
     5 FORMAT(1X,F5.1,1X 2F4.1,7(1X,F4.1),4(1X,F5.1))
       DO 40 A=1,10
          IF (R(A).GE.O.AND.SEX.EQ.1.0) THEN
              R(A)=R(A)**2
              TR(A)=TR(A)+R(A)
              TSR(A)=TSR(A)+R(A)**2
              TN(A)=TN(A)+1
          ENDIF
       IF (TN(A).LE.0) THEN
          XX=A-1
          GO TO 30
       ENDIF
    40 CONTINUE
    30 CONTINUE
       DO 50 B=1,XX
          TMR(B)=TR(B)/TN(B)
          IF (TN(B).EQ.1) TN(B)=1.0001
          VARR(B)=(TSR(B)-((TR(B)**2)/TN(B)))/(TN(B)-1)
    50 CONTINUE
       DO 60 C=1,XX
          WRITE (6,70) TN(C),TMR(C),VARR(C)
    70    FORMAT (3(2X,F12.3))
    60 CONTINUE
       TSUM=VARRP=0
       DO 80 BB=1,XX
       VARRP=VARRP+((TN(BB)-1)*VARR(BB))
       TSUM=TSUM+(TN(BB)-1)
    80 CONTINUE
       VARRP=VARRP/TSUM
    95 CONTINUE
       DLOOPS=DLOOPS+1
       DIFFER=TOTNUM=0
       DO 96 D=1,XX
```

```
         SUM(D)=TRSUM(D)=SUMSQ(D)=TNUM(D)=0
 96   CONTINUE
      DO 110 ZZ=1,1223
      READ (2,97,END=110)TL,SEX
 97   FORMAT (2X,F5.1,2X,F4.1)
      IF (SEX.EQ.2) GO TO 110
      SL=(TL*0.86)+1.56
      SL=SL**2
      DO 98 CC=1,XX
         V(CC)=SL*(TMR(CC)/VARRP)-(0.5*(TMR(CC)**2)/VARRP)
     B+ALOG(PROP(CC))
 98   CONTINUE
      AMAX=V(1)
      X=1
      DO 100 DD=2,XX
         IF (V(DD).GT.AMAX) THEN
         AMAX=V(DD)
         X=DD
      ENDIF
100   CONTINUE
      SUM(X)=SUM(X)+SL
      SUMSQ(X)=SUMSQ(X)+SL**2
      SL=SQRT(SL)
      TRSUM(X)=TRSUM(X)+SL
      TNUM(X)=TNUM(X)+1
110   CONTINUE
      WRITE (6,115)
115   FORMAT ('  AGE    NO.     SUM LEN.   MEAN LEN.    VARIANCE')
      DO 120 EE=1,XX
         TOTNUM=TOTNUM+TNUM(EE)
      IF (TNUM(EE).LE.0) THEN
         SLMEAN(EE)=-1.00
      ELSE
            SLMEAN(EE)=TRSUM(EE)/TNUM(EE)
      ENDIF
120   CONTINUE
      DO 135 GG=1,XX
         DIFFER=DIFFER+ABS((TNUM(GG)-TNUM1(GG)))
         TNUM1(GG)=TNUM(GG)
135   CONTINUE
      WRITE (6,140)
140   FORMAT ('  DIFFER   DLOOPS')
      WRITE (6,150) DIFFER,DLOOPS
150   FORMAT (2X,F6.1,4X,F5.1)
      VARRP=TSUM=0
      DO 160 FF=1,XX
         TR1(FF)=TR(FF)+SUM(FF)
         TSR1(FF)=TSR(FF)+SUMSQ(FF)
         TN1(FF)=TN(FF)+TNUM(FF)
         TMR(FF)=TR1(FF)/TN1(FF)
```

```
          VARR(FF)=(TSR1(FF)-((TR1(FF)**2)/TN1(FF)))/(TN1(FF)-1)
          VARRP=VARRP+((TN1(FF)-1)*VARR(FF))
          TSUM=TSUM+(TN1(FF)-1)
          WRITE (6,155) FF,INUM(FF),TRSUM(FF),SLMEAN(FF),VARR(FF)
  155     FORMAT (2X,F3.0,2X,F5.0,2X,F9.2,2X,F9.2,2X,F13.3)
  160 CONTINUE
      IF (DIFFER.LE.0.OR.DLOOPS.EQ.15) GO TO 1000
      VARRP=VARRP/TSUM
      REWIND 2
      GO TO 95
 1000 END
END OF FILE
```

APPENDIX 2 *O. Mossambicus* YIELD MODEL (ORCMOD)

(a) Terms used in ORCMOD

Term	Definition
ACCTH (N,M)	Monthly catch (t) by sex and age class computed from CATCH and DATFA.
AMORTR	Annual angling target yield as proportion of population susceptible to angling (≥1+).
ANGMORT	Target yield from angling (t).
BIOM (N)	Total mass of age classes N = 1,6.
BIOMANG	Mass of population susceptible to angling (t).
BIOMC	Population biomass (t).
BIOMCFI	Mass of population susceptible to gill net fishery (≥2+), (t).
BREEDF	Factor determining proportion of female population spawning in a given month.
CATCH (N)	Actual catch taken from age class (N), ($t.mth^{-1}$).
CATRAT	Sum of monthly commercial and angling target yields for each age class (t).
DATE	Month, January = 1 to December = 12.
DATFA	Factor to determine monthly yield from annual total.
DATREC	Factor to number 0+ month classes from start of breeding season.
DENS	Fish density ($g.m^{-3}$).
FEC (N)	Eggs produced per spawning per female of age class (N).
FISNO (N,M)	Number of fish by age class and sex.
FMORT	Target yield from commercial fishing (t).
FMORTR	Annual commercial target yield as proportion of population susceptible to commercial fishery.
HMORTO	Monthly density dependent survival of 0+ fish.
OOBIOMT	Total mass of 0+ age class (t).
OOFISH(N)	No. of 0+ fish in month class (N), N = 1,12.
OOFISHT	Total number of 0+ fish.
OOMASS(N)	Individual mass (g) of 0+ fish in month class (N).
OOMASST	Individual mass (g) of 0+ fish in month class (2).
PRINT	Output format instruction, 1 = full, 2 = summary.
RECRT	Total number of eggs produced per month.

Term	Definition
RECR1,6	Total number of eggs produced per month per age class, 1,6.
SLENG	Input file containing numbers and individual mass (g) per age class and sex.
SLM (N,M)	Standard length (cm) of each age class and sex.
STDLEN (N)	Standard length (cm) of each MONTH class of 0+ fish.
TEMP	Mean epilimnetic water temperature (°C)
TEMPFA	Factor to regulate monthly growth according to mean epilimnetic water temperature.
VOL	Volume of dam as a proportion of full supply volume.
WMAS (N,M)	Individual mass (g) of each age class and sex.
XMORTO	Proportion per month of age class suffering cold-induced winter mortality.
XSURVI5	Monthly expectation of natural death for fish in age classes 1-6.
YIELD	Realised annual yield (t).

Appendix 2b. The FORTRAN V Programme, ORCMOD.

```
      PROGRAM ORCMOD(TEMP,SLENG,UIT,TAPE1=TEMP,TAPE2=SLENG,TAPE6=UIT)
C     INITIAL VALUES. MASS IS IN KGS. THE YEAR RUNS FROM OCT TO SEPT
      DIMENSION OOFISH(12),OOBIOM(12),OOMASS(12),TO(12),FISNO(6,2)
     1,STDLEN(12),SLM(6,2),WMAS(6,2),BIOM(6),SLENG(6,6),CATCH(6)
     2,ACCTH(6,2),FEC(6),GRINC(6,2)
      INTEGER       IR,K,NR
      DOUBLE PRECISION  DSEED
C     *******
      ANGMORT=50.0
C     *******
      DSEED = 123567.0D
      DO 220 Y=1,10
      DO 8 KK=1,12
         OOFISH(KK)=OOBIOM(KK)=OOMASS(KK)=0
    8 CONTINUE
      OOMASST=0
      OOBIOMT=0
      RECRT=0
C     ********
      FMORT=0
C     ********
      BIOMC=319
C     ********
      PRINT=2
C     ********
      READ (2,16,END=1999) ((SLENG(I,J),J=1,6),I=1,6)
   16 FORMAT (6F10.0)
 1999 CONTINUE
      DO 17 II=1,6
         SLM(II,1)=SLENG(II,1)
         SLM(II,2)=SLENG(II,3)
         WMAS(II,1)=SLENG(II,2)
         WMAS(II,2)=SLENG(II,4)
         FISNO(II,1)=SLENG(II,5)
         FISNO(II,2)=SLENG(II,6)
   17 CONTINUE
      WRITE(6,9)
    9 FORMAT ('              ORCMOD.  MODEL TO PREDICT GROWTH AND'
     2,' PRODUCTION OF OREOCHROMIS MOSSAMBICUS IN HARTBEESPOORT DAM'//
     3 'MODEL RUNS FROM OCT TO SEPT.  DATA OUTPUT PER MONTH'/)
      WRITE(6,10)
   10 FORMAT ('  COMMERCIAL FISHING MORTALITY, FMORT, AND ANGLING '
     4,'MORTALITY,ANGMORT, TONNES REMOVED PER ANNUM ')
      WRITE(6,11)FMORT,ANGMORT
   11 FORMAT(3X,F8.1,3X,F8.1)
      WRITE(6,12)
   12 FORMAT ('  FISHING GEAR USED [ COMMERCIAL'
     1,'GILL-NETTING WITH 118MM & LARGER NETS]'/)
      WRITE(6,13)
   13 FORMAT ('   MN=MASS OF NTH AGE CLASS REMAINING [TONNES]'//)
```

```
      DO 200 Z=1,20
      TBIOMC=0
      DO 20 SS = 1,12
         OOMASS(SS)=0
         STDLEN(SS)=0
   20 CONTINUE
      OOMASST=0
      YIELD=0
      IF (PRINT.EQ.1) THEN
      WRITE(6,14)
   14 FORMAT ('     MONTH     M0       M1       M2       M3       M4'
     4,'       M5       M6')
      ENDIF
C     BIOMANG = THE BIOMASS OF THE POPULATION WHICH IS SUSCEPTIBLE
C     TO ANGLING (1+ FISH AND OLDER)
      BIOMCFI=BIOMANG=0
      DO 25 A=1,6
         BIOMANG=BIOMANG+((FISNO(A,1)*WMAS(A,1))+(FISNO(A,2)*
     1   WMAS(A,2)))/1.0E6
   25 CONTINUE
C     BIOMCFI = THE BIOMASS OF THE POPULATION WHICH WOULD BE
C     SUSCEPTIBLE TO A GILL NET FISHERY (2+ FISH AND OLDER)
      DO 30 B=4,6
         BIOMCFI=BIOMCFI+((FISNO(B,1)*WMAS(B,1))+(FISNO(B,2)*
     2   WMAS(B,2)))/1.0E6
   30 CONTINUE
      IF (BIOMANG.LE.0) THEN
         AMORTR=0
      ELSE
         AMORTR=ANGMORT/BIOMANG
      ENDIF
      IF (AMORTR.GT.1) AMORTR=1
      IF (BIOMCFI.LE.0) THEN
         FMORTR=0
      ELSE
         FMORTR=FMORT/BIOMCFI
      ENDIF
      IF (FMORTR.GT.1) FMORTR=1
      DO 33 G=1,6
         IF (G.EQ.1) CATRAT=AMORTR
         IF (G.GT.1) CATRAT=AMORTR+FMORTR
         CATCH(G)=CATRAT*(FISNO(G,1)+FISNO(G,2))
   33 CONTINUE
      DO 100 I=1,12
      DO 34 II=1,6
         GRINC(II,1)=(12.68-0.34*SLM(II,1))/12
         GRINC(II,2)=(12.40-0.40*SLM(II,2))/12
   34 CONTINUE
      OOINC=12.68/12
      READ(1,15,END=9999) DATE,TEMP,DATFA,DATREC,VOL
```

```
   15 FORMAT (2X,5F10.0)
 9999 TEMPFA=(0.0086*(TEMP**2.0783))/(0.0086*(17.75**2.0783))
      TEMPFA=TEMPFA*(12/13.8)
      IF (TEMP.LE.18) TEMPFA=0
C     XMORTO = JUVENILE (0+) WINTER MORTALITY
      K = 4
      NR = 1
      IR = 1
       IF (DATE.EQ.7) THEN
         CALL GGUD (DSEED,K,NR,IR)
         MORT=IR
       ENDIF
      IF (IR.EQ.1) XMORTO=0.0
      IF (IR.EQ.2) XMORTO=0.25
      IF (IR.EQ.3) XMORTO=0.75
      IF (IR.EQ.4) XMORTO=1.0
C     OOFISH = THE NUMBER OF FISH SURVIVING FROM THE 0 YEAR CLASS. BIOMO
C     IS THE BIOMASS OF THIS CLASS OBTAINED FROM OOFISH*GRWO.
C     XSURV15 IS THE NATURAL MORTALITY RATE FOR ALL FISH IN CLASSES 1-5
      XSURV15 = 1-(10**(ALOG10(0.54)/12))
C     HMORTO IS THE SURVIVAL OF 0+ FISH. THIS SURVIVAL IS
C     LINKED TO THE O.MOSS DENSITY (G/M3) AND IS THUS DENSITY
C     DEPENDENT
C     DATFA IS A CONVERSION FACTOR TO APPLY SEASONAL BIAS TO FISHING
C     MORTALITY (FMORT)
C     FEMALES GT 100G ARE CONSIDERED TO BE CAPABLE OF BREEDING
      DO 35 C=1,6
         DO 40 D=1,2
      IF (YIELD.LT.(ANGMORT+FMORT)) THEN
         ACCTH(C,D)=(CATCH(C)/2)*DATFA
         IF (ACCTH(C,D).GT.FISNO(C,D)) ACCTH(C,D)=FISNO(C,D)
         ACCTH(C,D)=ACCTH(C,D)*WMAS(C,D)/1.0E6
         YIELD=YIELD+ACCTH(C,D)
         FISNO(C,D)=FISNO(C,D)-(CATCH(C)/2)*DATFA
         IF (FISNO(C,D).LT.0) FISNO(C,D)=0
      ELSE
         ACCTH(C,D)=0
      ENDIF
         FISNO(C,D)=FISNO(C,D)*(1-XSURV15)
   40 CONTINUE
   35 CONTINUE
      SLM(1,1)=SLM(1,1)+(GRINC(1,1)*TEMPFA)
      WMAS(1,1)=SLMAS(SLM(1,1))
      SLM(1,2)=SLM(1,2)+(GRINC(1,2)*TEMPFA)
      WMAS(1,2)=SLMAS(SLM(1,2))
      FEC(1)=40.5*(WMAS(1,2)**0.63)
      IF (WMAS(1,2).LT.100) FEC(1)=0
C     BREEDF IS A FACTOR FOR DISTRIBUTING TOTAL FECUNDITY OVER THE
C     NORMAL BREEDING SEASON
      BREEDF=0
```

```
      IF (DATREC.GE.1.AND.DATREC.LE.5) BREEDF=0.49
      IF (DATREC.EQ.6) BREEDF=0.25
      RECR1=FEC(1)*BREEDF*FISNO(1,2)
C
C     THE SAME PROCEDURE IS REPEATED FOR SUCCESSIVE YEAR CLASSES.
C     THE SECOND YEAR CLASS
C
      SLM(2,1)=SLM(2,1)+(GRINC(2,1)*TEMPFA)
      WMAS(2,1)=SLMAS(SLM(2,1))
      SLM(2,2)=SLM(2,2)+(GRINC(2,2)*TEMPFA)
      WMAS(2,2)=SLMAS(SLM(2,2))
      FEC(2)=40.5*(WMAS(2,2)**0.63)
      RECR2=FEC(2)*BREEDF*FISNO(2,2)
C
C     THE THIRD YEAR CLASS
C
      SLM(3,1)=SLM(3,1)+(GRINC(3,1)*TEMPFA)
      WMAS(3,1)=SLMAS(SLM(3,1))
      SLM(3,2)=SLM(3,2)+(GRINC(3,2)*TEMPFA)
      WMAS(3,2)=SLMAS(SLM(3,2))
      FEC(3)=40.5*(WMAS(3,2)**0.63)
      RECR3=FEC(3)*BREEDF*FISNO(3,2)
      IF (RECR3.LT.0) RECR3=0
C
C     THE FOURTH YEAR CLASS
C
      SLM(4,1)=SLM(4,1)+(GRINC(4,1)*TEMPFA)
      WMAS(4,1)=SLMAS(SLM(4,1))
      SLM(4,2)=SLM(4,2)+(GRINC(4,2)*TEMPFA)
      WMAS(4,2)=SLMAS(SLM(4,2))
      FEC(4)=40.5*(WMAS(4,2)**0.63)
      RECR4=FEC(4)*BREEDF*FISNO(4,2)
      IF (RECR4.LT.0) RECR4=0
C
C     THE FIFTH YEAR CLASS
C
      SLM(5,1)=SLM(5,1)+(GRINC(5,1)*TEMPFA)
      WMAS(5,1)=SLMAS(SLM(5,1))
      SLM(5,2)=SLM(5,2)+(GRINC(5,2)*TEMPFA)
      WMAS(5,2)=SLMAS(SLM(5,2))
      FEC(5)=40.5*(WMAS(5,2)**0.63)
      RECR5=FEC(5)*BREEDF*FISNO(5,2)
      IF (RECR5.LT.0) RECR5=0
C
C     THE SIXTH YEAR CLASS
C
      SLM(6,1)=SLM(6,1)+(GRINC(6,1)*TEMPFA)
      WMAS(6,1)=SLMAS(SLM(6,1))
      SLM(6,2)=SLM(6,2)+(GRINC(6,2)*TEMPFA)
      WMAS(6,2)=SLMAS(SLM(6,2))
```

```
      FEC(6)=40.5*(WMAS(6,2)**0.63)
      RECR6=FEC(6)*BREEDF*FISNO(6,2)
      IF (RECR6.LT.0) RECR6=0
C
C     THE 0+ YEAR CLASS
C
      RECRT=RECR1+RECR2+RECR3+RECR4+RECR5+RECR6
      L=DATREC
      OOFISH(L)=RECRT
      OOBIOMT=0
      OOFISHT=0
      DO 70 J=1,L
         STDLEN(J)=STDLEN(J)+(OOINC*TEMPFA)
         OOMASS(J)=0.028*(STDLEN(J)**3.12)
         DENS=BIOMC/(192.8*VOL)
         HMORTO=1/(1+(DENS/2.440)**2.76)
         IF (STDLEN(J).GT.10.0) HMORTO=(1-XSURV15)
         IF (HMORTO.LT.0) HMORTO=0
         OOFISH(J)=(OOFISH(J)*(1.0-XMORTO)*HMORTO)
         OOFISHT=OOFISHT+OOFISH(J)
         OOBIOMT=OOBIOMT+(OOFISH(J)*OOMASS(J))
   70 CONTINUE
      OOMASST=OOMASS(2)
      OOBIOMT=OOBIOMT/1.0E6
      BIOMC=OOBIOMT
      DO 45 D=1,6
         BIOM(D)=((FISNO(D,1)*WMAS(D,1))+(FISNO(D,2)*WMAS(D,2)))
     5   /1.0E6
         BIOMC=BIOMC+BIOM(D)
   45 CONTINUE
      IF (PRINT.EQ.1) THEN
      WRITE(6,90)DATE,OOBIOMT,BIOM(1),BIOM(2),BIOM(3),BIOM(4),
     6BIOM(5),BIOM(6)
   90 FORMAT(2X,8(F8.0))
      ENDIF
      TBIOMC=TBIOMC+BIOMC
  100 CONTINUE
      REWIND 1
      TBIOMC=TBIOMC/12.0
      WRITE (6,91) TBIOMC
   91 FORMAT ('TOTAL BIOMASS = ',F8.2)
      WRITE (6,92) YIELD,MORT
   92 FORMAT ('  ANNUAL YIELD = ',F8.2,'  WINTER MORT.',I2)
      DO 115 E=6,2,-1
         DO 120 F=1,2
            SLM(E,F)=SLM(E-1,F)
            WMAS(E,F)=WMAS(E-1,F)
            FISNO(E,F)=FISNO(E-1,F)
  120    CONTINUE
  115 CONTINUE
```

```
      FISNO(1,1)=FISNO(1,2)=OOFISHT/2
      DO 125 LL=1,12
         IF (OOFISH(LL).GT.0) THEN
            WMAS(1,1)=WMAS(1,2)=OOMASS(LL+1)
            SLM(1,1)=SLM(1,2)=STDLEN(LL+1)
            GOTO 126
         ENDIF
  125 CONTINUE
  126 CONTINUE
      DO 18 MM = 1,12
      OOFISH(MM)=0
   18 CONTINUE
  200 CONTINUE
      REWIND 1
      REWIND 2
      ANGMORT=ANGMORT+0.0
      WRITE (6,210) ANGMORT
  210 FORMAT (////'     ANGMORT = ',F6.2)
  220 CONTINUE
      END
      FUNCTION SLMAS (STDL)
        SLMAS=0.031*(STDL**3.02)
      RETURN
      END
END OF FILE
```

APPENDIX 3. *C. Carpio* AND *C. gariepinus* YIELD MODELS (CYPMOD and CLAMOD)

(a) Terms used in both models.

Term	Definition
ADMORT	Total mortality of adults (≥1+). (t).
ASURV	Proportion of adult population which will survive one month.
BIOMS (N)	Mass of fish in each age class (N = 1,10), (t).
BMONTH	No. of months since spawning occurred. Spawning month = 1.
CATCH	Target yield per annum (t).
CLANG	Total mass of fish susceptible to exploitation (t).
COMP	Degree of compensation in 0+ density dependent survival relationship.
DENS	Density of population ($g.m^{-3}$).
DINCR (N)	Standard length growth increment of males ($cm.mth^{-1}$) in each age class.
FINCR (N)	Standard length growth increment of females ($cm.mth^{-1}$) in each age class.
FLAG	Flag to indicate spawning has already occurred in a given year.
FMORT (N)	No. of fish that die per month per age class.
PREWAT	Lake volume of previous month (% of full supply).
PRINT	Select printout format, 1 = full, 2 = summary.
SLNOS	Array of male and female numbers and standard length (cm) for each year class.
STIM	Change in lake volume over previous month (%).
SURV	0+ density dependent survival rate per month.
TBIOM	Total biomass (t).
THRESH	Threshold biomass above which density dependent effects dominate (t).
TMASS (N)	Mean mass per fish of age group (N), (g)
TYIELD	Realised yield per year (t).
YIELD (N)	Realised yield of each age class per month (no.).

(b) Terms used only in CYPMOD

ANGSEL (N)	Mass of age class N (t).
MONTH	Month, January = 1 to December = 12.
WATLEV	Lake volume as percentage of full supply.

(c) Terms used only in CLAMOD

ANGSEL (N)	Vulnerability to capture by angling according to mass.
ANGTOT	Sum of ANGSEL for all sex and age classes.
DAMLEV	Lake volume as percentage of full supply.
TIME	Months between spawning and end of model iteration year.
YMNTH	Month, January = 1 to December = 12.

Appendix 3d. The FORTRAN V Programme, CYPMOD.

```
      PROGRAM CYPMOD (CYPDAT,WATLEV,UITCYP,TAPE2=
     1CYPDAT,TAPE3=WATLEV,TAPE6=UITCYP)
C     THIS PROGRAM PREDICTS THE MONTHLY BIOMASS, BY AGE CLASSES, OF
C     CYPRINUS CARPIO IN HARTBEESPOORT DAM. THE INPUT DATA IS IN
C     TWO FILES. THE FIRST CONSISTS OF FEMALE AND MALE NUMBERS AND
C     MEAN SL PER AGE GROUP. THE SECOND CONTAINS MONTHLY DAM VOLUMES
C     WHICH CONTROL SPAWNING. GROWTH IS COMPUTED FROM THE REGRESSION
C     OF SL : INCREMENT.
      DIMENSION SLNOS(10,4),TMASS(10,2),BIOMS(10),FINCR(10),DINCR(10)
     1,FMORT(10),YIELD(10,2),ANGSEL(10)
C     **********
      CATCH=250.0
C     **********
      DO 400 AA=1,10
      MONTH=9
      BMONTH=9
      COMP=1.0
C     THE THRESHOLD BIOMASS IS CALCULATED WITH COMP AND SCALED TO
C     PRODUCE THE OBSERVED S=0.00000710 WHEN TBIOM=580 TONNES
      TBIOM=580.0
      DENS=580/(192.8*(100.0/100.0))
      THRESH = DENS/(10**(ALOG10((1/0.00000710)-1)/COMP))
      ASURV=10**(ALOG10(0.94)/12)
C     PRINT = 1 GIVES FULL MONTHLY PRINT OUT. = 2 GIVES ANNUAL FIGURES
C     ONLY
      PRINT = 2
      PREWAT=94
C     THE CATCH INPUT IS AN ESTIMATED ANNUAL YIELD.
C     AS THE STOCK IS DYNAMIC THIS WILL
C     NOT RESULT IN A YIELD EXACTLY EQUAL TO THE TARGET YIELD
      FMORT(1)=0
  199 CONTINUE
      READ (2,40,END=299) ((SLNOS(I,J),J=1,4),I=1,10)
   40 FORMAT (4F10.0)
  299 CONTINUE
          DO 60 J=1,4
          BIOMS(J)=0
   60 CONTINUE
      DO 300 KK=1,20
      WRITE (6,5) KK
    5 FORMAT ('  YEAR = ',I2/)
      ADMORT=0
      TYIELD=0
      TMBIO=0
      DO 10 J=1,10
          FINCR(J) = (29.67-(SLNOS(J,2)*0.50))/12
          IF (FINCR(J).LE.0) FINCR(J)=0
          DINCR(J) = (24.60-(SLNOS(J,4)*0.43))/12
          IF (DINCR(J).LE.0) DINCR(J)=0
   10  CONTINUE
```

```
      FLAG=1
      DO 200 K=1,12
      READ (3,50,END=399) WATLEV
   50 FORMAT (F10.0)
  399 CONTINUE
      IF (PRINT.EQ.1) THEN
      WRITE (6,11) MONTH
   11 FORMAT (//'  MONTH = ',I2)
      WRITE (6,15)
   15 FORMAT ('BIOMASS OF AGE CLASSES 0 TO 9 (TONNES) ')
      ENDIF
      CLANG = 0
      DO 110 N=1,10
         SLNOS(N,2)=SLNOS(N,2)+FINCR(N)
         SLNOS(N,4)=SLNOS(N,4)+DINCR(N)
         TMASS(N,1) = 0.037*(SLNOS(N,2)**2.94)
         TMASS(N,2) = 0.037*(SLNOS(N,4)**2.94)
  110 CONTINUE
      IF (TMASS(1,1).GE.150) THEN
         A = 1
      ELSE
         A = 2
         FMORT(1)=0
      ENDIF
      DO 12 L=A,10
         ANGSEL(L)=((SLNOS(L,1)*TMASS(L,1))+(SLNOS(L,3)*TMASS(L,2)))
     3   /1.0E6
         CLANG=CLANG+ANGSEL(L)
   12 CONTINUE
      DO 13 M = A,10
         FMORT(M)=(((CATCH/12)*ANGSEL(M)/CLANG)*1.0E6)/((TMASS(M,1)
     4   +TMASS(M,2))/2)
   13 CONTINUE
      DENS=TBIOM/(192.8*(WATLEV/100))
C     THE 0+ SURVIVAL RATE IS COMPUTED ACCORDING TO THE POSITION OF
C     TBIOM ON THE STOCK:0+ SURV CURVE .THE SHAPE OF THIS CURVE
C     DEPENDS ON THRESH AND COMP.
      IF (BMONTH.GE.1.AND.BMONTH.LE.2) THEN
         SURV = 1/(1+(DENS/THRESH)**COMP)
         SLNOS(1,1)=SLNOS(1,1)*(10**(ALOG10(SURV)/2))
      ELSEIF (BMONTH.GT.2.AND.TMASS(1,1).LT.150) THEN
         SLNOS(1,1)=SLNOS(1,1)*ASURV
      ENDIF
      SLNOS(1,3) = SLNOS(1,1)
C     MORTALITY IS DEDUCTED. FISHING MORTALITY IS FIRST REMOVED
C     AND TH N NATURAL MORTALITY AT A CONSTANT RATE OF 0.06 FOR
C     1+ AND OLDER FISH.
C     THE YIELD PER AGE CLASS PER SEX IS COMPUTED
      TBIOM=0.0
      DO 112 J = A,10
```

```
         IF (FMORT(J).GE.(SLNOS(J,1)+SLNOS(J,3))) FMORT(J)=
     9   (SLNOS(J,1)+SLNOS(J,3))
         SLNOS(J,1)=SLNOS(J,1)-(FMORT(J)/2)
         SLNOS(J,3)=SLNOS(J,3)-(FMORT(J)/2)
         TYIELD=TYIELD+(FMORT(J)*(TMASS(J,1)+TMASS(J,2))
     3)/2.0E06
C         NATURAL MORTALITY
         ADMORT=ADMORT+((SLNOS(J,1)*TMASS(J,1))+(SLNOS(J,3)*TMASS(J,2))
     1*(1-ASURV))
         ADMORT=ADMORT/1.0E06
         SLNOS(J,1)=SLNOS(J,1)*(ASURV)
         SLNOS(J,3)=SLNOS(J,3)*(ASURV)
  112    CONTINUE
         DO 111 II=1,10
         BIOMS(II)=(SLNOS(II,1)*TMASS(II,1))+(SLNOS(II,3)*TMASS(II,2))
         BIOMS(II)=BIOMS(II)/1.0E06
         TBIOM=TBIOM+BIOMS(II)
  111    CONTINUE
         IF (PRINT.EQ.1) THEN
         WRITE (6,130) (SLNOS(I,2),I=1,10)
  130    FORMAT (1X,10(1X,F6.2))
         WRITE (6,140) TBIOM
  140    FORMAT (/'  TOTAL BIOMASS = ',F10.2)
         ENDIF
         IF (MONTH.EQ.10) THEN
         DO 150 I=1,4
            DO 160 J=9,1,-1
            SLNOS(J+1,I)=SLNOS(J,I)
  160       CONTINUE
            SLNOS(1,I) = 0
  150    CONTINUE
         ENDIF
         STIM=WATLEV-PREWAT
C        RECRUITMENT TO THE 0+ CLASS IS COMPUTED FROM THE SUM OF THE
C        PRODUCTS OF MEAN FECUNDITY AND NUMBER OF FEMALES FOE EACH
C        AGE CLASS
         IF (STIM.GE.5.AND.FLAG.EQ.1) THEN
            DO 180 LL=2,10
            SLNOS(1,1)=SLNOS(1,1)+(SLNOS(LL,1)*(0.45*(SLNOS(LL,2)
     2       **3.58)))*0.5
            SLNOS(1,3)=SLNOS(1,1)
  180       CONTINUE
            SLNOS(1,2)=SLNOS(1,4)=0.0
            FLAG=2
            BMONTH=0
            IF (MONTH.GE.10) Z=10-(MONTH-12)
            IF (MONTH.LT.10) Z=10-MONTH
            FINCR(1)=27.8/Z
            DINCR(1)=27.6/Z
         ENDIF
```

```
      MONTH=MONTH+1
      PREWAT=WATLEV
      BMONTH=BMONTH+1
      IF (MONTH.EQ.13) MONTH=1
      IF (BMONTH.EQ.13) BMONTH=1
      TMBIO=TMBIO+TBIOM
  200 CONTINUE
      REWIND 3
      CLANG = 0
      TMBIO=TMBIO/12.0
      WRITE (6,210) ADMORT,TYIELD,TMBIO
  210 FORMAT (/'ANNUAL MORTALITY 1+ & OLDER = ',F8.0,/' TOTAL YIELD,'
     6,' = ',F8.0,/,'TOTAL BIOMASS = ',F8.0,//)
  300 CONTINUE
      REWIND 2
      CATCH=CATCH+25
  400 CONTINUE
      END
END OF FILE
```

Appendix 4. The FORTRAN V Programme, CLAMOD.

```
      PROGRAM CLAMOD (CLDAT2,DAMLEV,UITCL,TAPE2=
     1CLDAT2,TAPE3=DAMLEV,TAPE6=UITCL)
C     THIS PROGRAM PREDICTS THE MONTHLY BIOMASS, BY AGE CLASSES, OF
C     CLARIAS GARIEPINUS IN HARTBEESPOORT DAM. THE INPUT DATA IS IN
C     TWO FILES. THE FIRST CONSISTS OF FEMALE AND MALE NUMBERS AND
C     MEAN SL PER AGE GROUP. THE SECOND CONTAINS MONTHLY DAM VOLUMES
C     WHICH CONTROL SPAWNING. GROWTH IS COMPUTED FROM THE REGRESSION
C     OF SL : INCREMENT.
      DIMENSION SLNOS(10,4),TMASS(10,2),BIOMS(10),FINCR(10),DINCR(10)
     1,FMORT(10,2),YIELD(10,2),ANGSEL(10,2)
C     **********
      CATCH = 10
C     **********
      DO 400 AA=1,10
      YMNTH=9.0
      TIME=12
      BMONTH=9
      COMP=1.0
C     THE THRESHOLD BIOMASS IS CALCULATED WITH COMP AND SCALED TO
C     PRODUCE THE OBSERVED S=0.00000252 WHEN TBIOM=290 TONNES
      TBIOM=290.0
      DENS=290.0/(192.8*(100.0/100.0))
      THRESH = DENS/(10**(ALOG10((1.0/0.00000252)-1.0)/COMP))
      ASURV=10**(ALOG10(0.91)/12.0)
C     PRINT = 1 GIVES FULL MONTHLY PRINT OUT. = 2 GIVES ANNUAL FIGURES
C     ONLY
      PRINT = 2
      PREWAT=100.0
C     THE CATCH INPUT IS AN ESTIMATED ANNUAL YIELD.
C     AS THE STOCK IS DYNAMIC THIS WILL
C     NOT RESULT IN A YIELD EXACTLY EQUAL TO THE TARGET YIELD
      FMORT(1,1)=FMORT(1,2)=0
  199 CONTINUE
      READ (2,40,END=299) ((SLNOS(I,J),J=1,4),I=1,10)
   40 FORMAT (4F10.0)
  299 CONTINUE
         DO 60 J=1,4
         BIOMS(J)=0
   60 CONTINUE
      DO 300 KK=1,20
      WRITE (6,5) KK
    5 FORMAT (/'  YEAR = ',I2)
      ADMORT=0
      TYIELD=0
      TMBIO=0
      FINCR(1)=24.6/TIME
      DINCR(1)=21.9/TIME
      DO 10 J=2,10
         FINCR(J) = (16.11-(SLNOS(J,2)*0.21))/12.0
         IF (FINCR(J).LE.0) FINCR(J)=0
```

```
      DINCR(J) = (18.39-(SLNOS(J,4)*0.19))/12.0
      IF (DINCR(J).LE.0) DINCR(J)=0
 10   CONTINUE
      FLAG=1
      DO 200 K=1,12
      READ (3,50,END=399) DAMLEV
 50   FORMAT (F10.0)
 399  CONTINUE
      IF (PRINT.EQ.1) THEN
      WRITE (6,11) YMNTH
 11   FORMAT (//'  MONTH = ',F4.1)
      WRITE (6,15)
 15   FORMAT ('BIOMASS OF AGE CLASSES 0 TO 9 (TONNES) ')
      ENDIF
C     THE 0+ SURVIVAL RATE IS COMPUTED ACCORDING TO THE POSITION OF
C     TBIOM ON THE STOCK:0+ SURV CURVE .THE SHAPE OF THIS CURVE
C     DEPENDS ON THRESH AND COMP.
      DENS=TBIOM/(192.8*(DAMLEV/100.0))
      IF (BMONTH.GE.1.AND.BMONTH.LE.3) THEN
         SURV = 1.0/(1+(DENS/THRESH)**COMP)
         SLNOS(1,1)=SLNOS(1,1)*(10**(ALOG10(SURV)/3.0))
      ELSE
         SLNOS(1,1)=SLNOS(1,1)*0.992
      ENDIF
      SLNOS(1,3) = SLNOS(1,1)
      DO 110 N=1,10
         SLNOS(N,2)=SLNOS(N,2)+FINCR(N)
         SLNOS(N,4)=SLNOS(N,4)+DINCR(N)
         TMASS(N,1) = 0.015*(SLNOS(N,2)**2.96)
         TMASS(N,2) = 0.015*(SLNOS(N,4)**2.96)
 110  CONTINUE
C     MORTALITY IS DEDUCTED. FISHING MORTALITY IS FIRST REMOVED
C     AND THEN NATURAL MORTALITY AT A CONSTANT RATE OF 0.09 FOR
C     1+ AND OLDER FISH.
      ANGTOT=0.0
      DO 25 I=1,10
         ANGSEL(I,1) = 3.35*(TMASS(I,1)**1.01)
         ANGSEL(I,2) = 3.35*(TMASS(I,2)**1.01)
         ANGSEL(1,1)=ANGSEL(1,2)=ANGSEL(2,1)=ANGSEL(2,2)=0
         ANGTOT=ANGTOT+ANGSEL(I,1)+ANGSEL(I,2)
         FMORT(I,1)=FMORT(I,2)=0
 25   CONTINUE
C     THE YIELD PER AGE CLASS PER SEX IS COMPUTED
      DO 90 L=10,2,-1
         YIELD(L,1)=(CATCH*(ANGSEL(L,1)/ANGTOT))/12.0
         YIELD(L,2)=(CATCH*(ANGSEL(L,2)/ANGTOT))/12.0
 90   CONTINUE
      TBIOM=0.0
      DO 112 J = 10,2,-1
      FMORT(J,1)=FMORT(J,1)+YIELD(J,1)/(TMASS(J,1)/1.0E06)
```

```
      FMORT(J,2)=FMORT(J,2)+YIELD(J,2)/(TMASS(J,2)/1.0E06)
      IF (FMORT(J,1).GT.SLNOS(J,1)) THEN
         FMORT(J-1,1)=FMORT(J-1,1)+(FMORT(J,1)-SLNOS(J,1))
     1   *(TMASS(J,1)/TMASS(J-1,1))
         FMORT(J,1)=SLNOS(J,1)
      ELSEIF (FMORT(J,2).GT.SLNOS(J,3)) THEN
         FMORT(J-1,2)=FMORT(J-1,2)+(FMORT(J,2)-SLNOS(J,3))
     2   *(TMASS(J,2)/TMASS(J-1,2))
         FMORT(J,2)=SLNOS(J,3)
      ENDIF
      SLNOS(J,1)=SLNOS(J,1)-FMORT(J,1)
      SLNOS(J,3)=SLNOS(J,3)-FMORT(J,2)
      TYIELD=TYIELD+((FMORT(J,1)*TMASS(J,1))+(FMORT(J,2)*TMASS(J,2)
     3))/1.0E06
C       NATURAL MORTALITY
      ADMORT=ADMORT+((SLNOS(J,1)*TMASS(J,1))+(SLNOS(J,3)*TMASS(J,2))
     1*(1-ASURV))
      SLNOS(J,1)=SLNOS(J,1)*(ASURV)
      SLNOS(J,3)=SLNOS(J,3)*(ASURV)
  112 CONTINUE
      ADMORT=ADMORT/1.0E06
      DO 111 II=1,10
      BIOMS(II)=(SLNOS(II,1)*TMASS(II,1))+(SLNOS(II,3)*TMASS(II,2))
      BIOMS(II)=BIOMS(II)/1.0E06
      TBIOM=TBIOM+BIOMS(II)
  111 CONTINUE
      IF (PRINT.EQ.1) THEN
      WRITE (6,130) (BIOMS(I),I=1,10)
  130 FORMAT (1X,10(1X,F6.2))
      WRITE (6,140) TBIOM
  140 FORMAT (/'  TOTAL BIOMASS = ',F10.2)
      ENDIF
      IF (YMNTH.EQ.10.0) THEN
      DO 150 I=1,4
         DO 160 J=9,1,-1
         SLNOS(J+1,I)=SLNOS(J,I)
  160    CONTINUE
         SLNOS(1,I) = 0
  150 CONTINUE
      ENDIF
      STIM=DAMLEV-PREWAT
C     RECRUITMENT TO THE 0+ CLASS IS COMPUTED FROM THE SUM OF THE
C     PRODUCTS OF MEAN FECUNDITY AND NUMBER OF FEMALES FOR EACH
C     AGE CLASS
      IF (STIM.GE.5.AND.FLAG.EQ.1) THEN
         DO 180 LL=3,10
         SLNOS(1,1)=SLNOS(1,1)+(SLNOS(LL,1)*((0.17*SLNOS(LL,2)
     2   **3.30))*0.5)
         SLNOS(1,3)=SLNOS(1,1)
  180    CONTINUE
```

```
          SLNOS(1,2)=SLNOS(1,4)=0.0
          FLAG=2
          BMONTH=0
          IF (YMNTH.GE.10.0) TIME=((12.0-YMNTH)+10.0)
          IF (YMNTH.LT.9.0) TIME=(10.0-YMNTH)
          IF (YMNTH.EQ.9.0) TIME=12
      ENDIF
      YMNTH=YMNTH+1
      PREWAT=DAMLEV
      BMONTH=BMONTH+1
      IF (YMNTH.EQ.13.0) YMNTH=1.0
      IF (BMONTH.EQ.13) BMONTH=1
      TMBIO=TMBIO+TBIOM
  200 CONTINUE
      REWIND 3
      TMBIO=TMBIO/12.0
      WRITE (6,210) TYIELD,TMBIO
  210 FORMAT (' TOTAL YIELD,'
     6,' = ',F8.0,'TOTAL BIOMASS = ',F8.0,)
  300 CONTINUE
      REWIND 2
      CATCH=CATCH+5
  400 CONTINUE
      END
END OF FILE
```

APPENDIX 5. MULTI-SPECIES PRODUCTION MODEL (MULTI)

(a) Terms used in MULTI

For variables Cn, On and Sn, n = 1 - 4 where

1 = *O. mossambicus*
2 = *C. carpio*
3 = *C. gariepinus*
4 = *predator*

Term	Definition
Cn	Realised annual catch of species n (t)
D1	Weighted sum of previous five years' winter mortality factor
H	Surface area of the lake (ha)
On	Running total of realised annual catches of species n (t)
P	Mean lake total phosphorus concentration ($\mu g.\ell^{-1}$)
P1	Mean lake pH
Sn	Potential annual yield of species n (t)
T	Total annual yield incorporating influence of a predator
W(K)	Winter mortality factor on a scale of 1 to 4 for K previous years where K = 1 - 5
Y	Total annual yield predicted from phosphorus concentration.

APPENDIX 5. MULTI-SPECIES PRODUCTION MODEL (MULTI)

(a) Terms used in MULTI

For variables Cn, On and Sn, n = 1 - 4 where

1 = *C. mossambicus*
2 = *C. carpio*
3 = *C. gariepinus*
4 = *predator*

Term	Definition
Cn	Realised annual catch of species n (t)
D1	Weighted sum of previous five years' winter mortality factor
H	Surface area of the lake (ha)
On	Running total of realised annual catches of species n (t)
P	Mean lake total phosphorus concentration ($\mu g.\ell^{-1}$)
P1	Mean lake pH
Sn	Potential annual yield of species n (t)
T	Total annual yield incorporating influence of a predator
W(K)	Winter mortality factor on a scale of 1 to 4 for K previous years where K = 1 - 5
Y	Total annual yield predicted from phosphorus concentration.

Appendix 5b. The BASIC Programme, MULTI.

```
10 PRINT "MULTI, A MODEL TO PREDICT YIELD OF DIFFERENT"
20 PRINT "SPECIES IN A SOUTHERN AFRICAN HIGHVELD INPOUNDMENT"
40 DIM W(5)
41 LET O1=0
42 LET O2=0
43 LET O3=0
45 PRINT "ENTER TOTAL PHOSPHORUS CONCENTRATION UG/L"
50 INPUT P
60 PRINT "ENTER SURFACE AREA OF LAKE HA"
70 INPUT H
80 LET Y=26.0*(10**((LGT(P)-LGT(1.55))/0.82))**0.60
90 LET Y=(10**(-6.0+(2.0*LGT(Y))))*(H*1.0E04)*10.0
100 LET Y=Y/1.0E06
110 LET P1=7.71+0.0017*P
120 IF P1<8.5 THEN 140
125 IF P1>11.0 THEN 1500
130 GOTO 200
140 PRINT "IS THERE A PREDATORY FISH IN THE SYSTEM?"
150 PRINT "YES=1, NO=2"
160 INPUT Y1
170 IF Y1=2 THEN 200
180 LET S4=0.05
200 PRINT "DO YOU WISH TO REMOVE OREOCHROMIS? Y=1,N=2"
210 INPUT Y1
220 IF Y1=1 THEN 300
240 PRINT "DOES THE WATER TEMPERATURE FALL BELOW 14 DEG  Y=1,N=2"
250 INPUT Y2
260 IF Y2=1 THEN 290
270 LET S1=0.53
280 GOTO 310
290 LET S1=0.53
295 LET F1=1
297 GOTO 310
300 S1=0.0
310 PRINT "DO YOU WISH TO REMOVE CYPRINUS? Y=1,N=2"
320 INPUT Y3
330 IF Y3=1 THEN LET S2=0.0
340 IF Y3=2 THEN LET S2=0.44
350 PRINT "DO YOU WISH TO REMOVE CLARIAS? Y=1,N=2"
355 INPUT Y4
360 IF Y4=1 THEN LET S3=0.0
370 IF IF WINTER MORTALITY OCCURS, THE ANNUAL YIELD OF OREOCHROMIS
445 REM IS RANDOM AND THE PROGRAM CALLS THE RANDOM FUNCTION
```

```
447 IF F1=1 THEN 460
450 GOTO 600
460 FOR I=1 TO 5
470 LET W(I)=INT(4*RND(-1)+1)
480 NEXT I
490 LET D1=0
500 GOTO 1000
600 FOR I=1 TO 20
620 IF F1=1 THEN 1100
625 LET C1=S1
630 LET C2=S2+(S1-C1)*(S2/(S2+S3))
640 LET C3=S3+(S1-C1)*(S3/(S2+S3))
645 PRINT "YEAR     ";I
650 PRINT "TOTAL YIELD ="; (T+S4)
655 PRINT "PREDATOR YIELD ="; C4
660 PRINT "OREOCHROMIS YIELD ="; C1
670 PRINT "CYPRINUS YIELD ="; C2
680 PRINT "CLARIAS YIELD ="; C3
685 PRINT "     "
686 O1=O1+C1
687 O2=O2+C2
688 O3=O3+C3
690 NEXT I
691 LET O1=O1/20
692 PRINT O1
693 LET O2=O2/20
694 PRINT O2
695 LET O3=O3/20
696 PRINT O3
700 GOTO 1530
1000 FOR K=1 TO 5
1010 LET D1=D1+W(K)
1020 NEXT K
1040 GOTO 600
1100 D1=(W(1)*0.5)+W(2)+W(3)+(W(4)*0.5)+(W(5)*0.25)
1105 LET C1=(S1+((S1/9.75)*3.25))-(D1*S1/9.75)
1107 IF C1<0.0 THEN LET C1=0.0
1110 FOR L=5 TO 2 STEP -1
1120 LET W(L)=W(L-1)
1130 NEXT L
1150 LET W(1)=INT(4*RND(-1)+1)
1160 D1=0
1170 FOR M=1 TO 5
1180 D1=D1+W(M)
1190 NEXT M
1200 GOTO 630
1500 PRINT "THE PH OF THE WATER IS ABOVE 11.0. CLARIAS MAY SURVIVE"
1510 PRINT "BUT THE OTHER SPECIES WOULD PROBABLY DIE OUT."
1530 END
END OF FILE
```

Appendix 5c. Examples of the use of MULTI to demonstrate input required and interaction between the model and the user. Output from the first year is also shown. A = run with 500 μg.ℓ$^{-1}$ phosphorus. B = run with 100 μg.ℓ$^{-1}$ phosphorus at which the pH is low enough for survival of the predator.

A. MULTI, a model to predict yield of different species in a southern African highveld impoundment.

Enter total phosphorus concentration μg/ℓ$^{-1}$

? 500

Enter surface area of lake ha

? 2000

Do you wish to remove *Oreochromis*? y = 1, n = 2

? 2

Does the water temperature fall below 14° y = 1, n = 2

? 1

Do you wish to remove *Cyprinus*? y = 1, n = 2

? 2

Do you wish to remove *Clarias*? y = 1, n = 2

? 2

year 1

Total yield = 634.096

Predator yield = 0

Oreochromis yield = 120.641

Cyprinus yield = 480.681

Clarias yield = 32.7737